```
(%i1) kill(all)$ load(draw)$
```

```
(%i2) f(%sigma,r, R) := (tanh(%sigma*(r+R)) - tanh(%sigma*(r-R)))/(2*tanh(%sigma*R));
```

(%o2) $f\left(\sigma , r , R\right) := \dfrac{\tanh\left(\sigma \left(r+R\right)\right)-\tanh\left(\sigma \left(r-R\right)\right)}{2 \tanh\left(\sigma R\right)}$

The following graph shows the tophat nature of the f function that multiplies the velocity in the ß displacement element.

First, make a list of f(%sigma,r, R) function for several values of sigma (set the ship dimension R = 1):

```
(%i3) flist: makelist(f(%sigma,r, 1), %sigma, [1, 5, 10, 15, 20, 23])$
```

And plot them all:

```
(%i4) wxplot2d(flist,[r,-2,2],[xlabel,"r"],[ylabel,"Warp Field (sigma = 1..23)"],[legend, false]);
```

(%t4)



(%o4)

More useful, perhaps, is a 3d surface plot:

```
(%i5) wxdraw3d(terminal       = png,
               xlabel ="r", ylabel = "sigma",
               wired_surface  = true,
               enhanced3d     = true,
               colorbox       = false,
               contour_levels = 16,
               contour        = base,
               explicit(f(%sigma,r,1),r,-2,2,%sigma,1,23));
```

(%t5)



(%o5)

Now for the theta function. For convenience, let's define pieces of the eventual function. First, define

```
(%i6) Z(x,%rho,xs) := sqrt(%rho^2+(x-xs)^2);
```

(%o6) $Z\left(x, \rho, xs\right) := \sqrt{\rho^2 + \left(x - xs\right)^2}$

The first factor of the theta function can then be written as

```
(%i7) f1(x,%rho,xs) := (x-xs)/Z(x,%rho,xs);
```

(%o7) $f1\left(x, \rho, xs\right) := \frac{x - xs}{Z\left(x, \rho, xs\right)}$

```
(%i8) f1(x,%rho,xs);
```

(%o8) $\frac{x - xs}{\sqrt{\left(x - xs\right)^2 + \rho^2}}$

The second factor is just the derivative of f with respect to r:

```
(%i9) f2(x,%rho,%sigma,xs) := diff(f(%sigma,r, R),r);
```

(%o9) $f2\left(x, \rho, \sigma, xs\right) := diff\left(f\left(\sigma, r, R\right), r\right)$

```
(%i10) f2(x,%rho,%sigma,xs);
```

(%o10) $\frac{\sigma\ \mathrm{sech}\left(\sigma\ \left(R + r\right)\right)^2 - \sigma\ \mathrm{sech}\left(\sigma\ \left(r - R\right)\right)^2}{2\ \tanh\left(\sigma\ R\right)}$

except that we substitute the Z we defined above, for r, into this factor f2:

```
(%i11) f2s(x,%rho,%sigma,xs) := subst(Z(x,%rho,xs),r,f2(x,%rho,%sigma, xs));
```

(%o11) $\mathrm{f2s}\left(x,\rho,\sigma,xs\right):=\mathrm{subst}\left(\mathrm{Z}\left(x,\rho,xs\right),r,\mathrm{f2}\left(x,\rho,\sigma,xs\right)\right)$

```
(%i12) f2s(x,%rho,s,xs);
```

(%o12)
$$\frac{s\,\mathrm{sech}\left(s\left(R+\sqrt{\left(x-xs\right)^2+\rho^2}\right)\right)^2-s\,\mathrm{sech}\left(s\left(\sqrt{\left(x-xs\right)^2+\rho^2}-R\right)\right)^2}{2\tanh\left(s\,R\right)}$$

The theta function is thus

```
(%i13) %theta(x,%rho,%sigma,xs) := f1(x,%rho,xs)*f2s(x,%rho,%sigma, xs);
```

(%o13) $\theta\left(x,\rho,\sigma,xs\right):=\mathrm{f1}\left(x,\rho,xs\right)\,\mathrm{f2s}\left(x,\rho,\sigma,xs\right)$

```
(%i14) %theta(x,%rho,%sigma,xs);
```

(%o14)
$$\frac{\left(x-xs\right)\left(\sigma\,\mathrm{sech}\left(\sigma\left(R+\sqrt{\left(x-xs\right)^2+\rho^2}\right)\right)^2-\sigma\,\mathrm{sech}\left(\sigma\left(\sqrt{\left(x-xs\right)^2+\rho^2}-R\right)\right)^2\right)}{2\sqrt{\left(x-xs\right)^2+\rho^2}\,\tanh\left(\sigma\,R\right)}$$

And now we can plot the theta function. Again, set the ship dimension R = 1:

```
(%i15) R: 1;
```

(%o15) 1

```
(%i16) wxdraw3d(terminal       = png,
                xlabel ="x", ylabel="rho",
                wired_surface  = true,
                enhanced3d     = true,
                colorbox       = false,
                contour_levels = 16,
                contour        = base,
                explicit(%theta(x,%rho,8,27),x,25.5,28.5,%rho,-1.5,1.5));
```

(%t16)



(%o16)